N E
12 14

NASA CONTRACTOR REPORT

NASA CR-129005

MULTIVARIATE NORMAL INTEGRATION

By Michael C. Carter
Appalachian State University
Department of Mathematical Sciences
Boone, North Carolina 28607

May 24, 1973

Final

Prepared for

NASA-GEORGE C. MARSHALL SPACE FLIGHT CENTER
Marshall Space Flight Center, Alabama 35812

(NASA-CR-129005) MULTIVARIATE NORMAL INTEGRATION Final Report (Appalachian State Univ.) 40 p

N77-70521

Unclas
00/96 09757

TECHNICAL REPORT STANDARD TITLE PAGE

1. REPORT NO. NASA CR-129005	2. GOVERNMENT ACCESSION NO.	3. RECIPIENT'S CATALOG NO.
4. TITLE AND SUBTITLE MULTIVARIATE NORMAL INTEGRATION		5. REPORT DATE May 24, 1973
		6. PERFORMING ORGANIZATION CODE
7. AUTHOR(S) Michael C. Carter		8. PERFORMING ORGANIZATION REPORT # Technical Report No. 4
9. PERFORMING ORGANIZATION NAME AND ADDRESS Appalachian State University, Department of Mathematical Sciences, Boone, North Carolina 28607		10. WORK UNIT NO.
		11. CONTRACT OR GRANT NO. NAS 8-29286
12. SPONSORING AGENCY NAME AND ADDRESS National Aeronautics and Space Administration, Washington, D. C.		13. TYPE OF REPORT & PERIOD COVERED Contractor Final
		14. SPONSORING AGENCY CODE

15. SUPPLEMENTARY NOTES

Work performed for Aerospace Environment Division, Aero-Astrodynamics Laboratory, Marshall Space Flight Center, Alabama; Mr. L. W. Falls NASA Contract Monitor.

16. ABSTRACT

This paper presents a Monte Carlo approach to the evaluation of multivariate normal integrals over rectangular regions for dimensions less than 6 and over elliptical regions in the bivariate case. The general theory is briefly discussed and the formulas used in this paper are presented. Several examples of a meteorological nature are presented to explain the process and demonstrate the application of the computer program developed to perform the integrations mentioned above. Appended is the computer program listing and a description of input card formats.

17. KEY WORDS Multivariate Normal Distribution Statistical Analysis Monte Carlo Methods	18. DISTRIBUTION STATEMENT Unclassified-unlimited E. D. Geissler Director, Aero-Astrodynamics Laboratory

19. SECURITY CLASSIF. (of this report)	20. SECURITY CLASSIF. (of this page)	21. NO. OF PAGES	22. PRICE
Unclassified	Unclassified	40	NTIS

MSFC - Form 3292 (Rev December 1972)

For sale by National Technical Information Service, Springfield, Virginia 22151

1. Introduction

This paper presents the results of an investigation into the application of the Monte Carlo integration technique in the evaluation of multivariate normal (MVN) integrals. Specifically the theory and resultant computer program is developed to evaluate MVN integrals for the following situations:

1) Over rectangular areas for dimension $M \leq 5$ utilizing completely arbitrary input parameters (mean vector and variance-covariance matrix) and

2) Over elliptical regions for arbitrary bivariate normal densities.

Section 2 presents pertinent results on the MVN distribution and describes the Monte Carlo technique in general applications. Section 3 presents the theory pertinent to our particular applications and section 4 discusses specific applications. The appendix presents the computer program and discusses the preparation of input cards.

2. General Theory

The Monte Carlo technique is, in general, a random simulation of a deterministic process. In integral evaluation one numerically evaluates a quantity whose expected value is the value of the integral and we can apply those statistical techniques applicable to such procedures.

The general principles of Monte Carlo procedures are well documented. The texts by Shreider (1964) and Newman and Odell (1971) provide excellent discussions on the Monte Carlo technique of integration.

As a simple, and inefficient, example consider the integral

$$\theta = \int_0^1 x\, dx.$$

The graph of $x, 0 \le x \le 1$, is in the unit square making θ the probability that a point, selected at random from the unit square, lies below the line $f(x) = x$. Thus if we select N pairs of uniform random numbers, say, (u_i, v_i), $i = 1, N$, and if K is the number of points where $v_i < u_i$, then the ratio K/N estimates θ. This is the "shotgun" or "hit-or-miss" technique and it is the most efficient. Another famous example is the estimation of π in the Buffon Needle Problem. Let the notation $\underline{x}$ denote an arbitrary vector of length m and $f(x)$ a real valued function of $\underline{x}$. We will use the notation $\int_{R_m} \ldots d\underline{x}$ to replace the lengthy $\int_{-\infty}^{\infty} \ldots \int_{-\infty}^{\infty} \ldots dx_1 dx_2 \ldots dx_m$. Suppose we desire to evaluate the integral

$$\theta = \int_{R_m} f(\underline{x})\, g(\underline{x})\, d\underline{x}. \tag{1}$$

where $g(\underline{x})$ denotes a probability density function on R_m. The integral (1) is then merely the expected value of the function $f(\underline{x})$ and can be estimated by evaluating the quantity

$$\hat{\theta} = \frac{1}{N} \sum_{i=1}^{N} f(\underline{x}_i)$$

where $\underline{x}_i$, i=1, N, are samples from the pdf $g(\underline{x})$. We directly have

$$\mathrm{Var}(\hat{\theta}) = \frac{1}{N}\,\mathrm{Var}(f(\underline{x})) = \frac{1}{N}\int_{R_m}(f(\underline{x})-\theta)^2 g(\underline{x})\,d\underline{x}$$

which is estimated by the quantity

$$S^2 = \frac{1}{N-1}\sum_{i=1}^{n}(f(\underline{x}_i)-\hat{\theta})^2$$

giving an estimated standard error of $\hat{e} = S/\sqrt{N}$. This quantity can be used to put confidence intervals on θ by the standard method.

One useful way to reduce the magnitude of $\mathrm{Var}(\hat{\theta})$ and, consequently, S^2 is to "remove the regular part". Suppose there exists a function $h(\underline{x})$ on R_m that approximates $f(\underline{x})$ well on R_m and further suppose that the value

$$\Psi = \int_{R_m} h(\underline{x})\; d\underline{x}$$

is known. We then have

$$\theta = \Psi + \int_{R_m}(f(\underline{x})-h(\underline{x}))d\underline{x}. \qquad (2)$$

The variance of $f(\underline{x})-h(\underline{x})$, where $\underline{x}$ has pdf $g(\underline{x})$, is

$$\mathrm{Var}(f(\underline{x})-h(\underline{x})) = \mathrm{Var}(f(\underline{x})) + \mathrm{Var}(h(\underline{x})) - 2\,\mathrm{Cov}(f(\underline{x}),h(\underline{x}))$$

and if $\mathrm{Var}(h(\underline{x})) < 2\,\mathrm{Cov}(f(\underline{x}),\, h(\underline{x}))$ we have immediately the result that

$$\mathrm{Var}(f(\underline{x})-h(\underline{x})) < \mathrm{Var}(f(\underline{x})). \qquad (3)$$

The MVN distribution is well documented and this discussion is merely to introduce notation and present results to be used in section 3.

The MVN density is the expression

$$f(\underline{x}|\underline{\mu},\Sigma) = \frac{1}{(2\Pi)^{\frac{m}{2}}|\Sigma|^{1/2}} e^{-1/2(\underline{x}-\underline{\mu})'\Sigma^{-1}(\underline{x}-\underline{\mu})}$$

where $\underline{\mu}$ is the vector of means and Σ is the positive definite, symmetric variance-covariance matrix. For such matrices there always exists a lower triangular matrix A such that $\Sigma = AA'$. Immediately we have $|\Sigma| = |A||A'| = |A'|^2$ giving $|\Sigma|^{1/2} = |A'|$ and, as A' is upper triangular, $|A'|$ is the product of the diagonal elements of A. If we define a linear transformation $\underline{y} = (A^{-1})'(\underline{x}-\underline{\mu})$ then $\underline{y}'\underline{y} = (\underline{x}-\underline{\mu})'A^{-1}(A^{-1})'(\underline{x}-\underline{\mu}) = (\underline{x}-\mu)'\Sigma^{-1}(\underline{x}-\underline{\mu})$. This result yields the equivalent pdf

$$g(\underline{y}) = \frac{1}{(2\Pi)^{\frac{m}{2}}} e^{-1/2\Sigma y_i^2} \quad .$$

Thus by a suitable linear transformation any MVN pdf can be reduced to the product of independent standard normal variates. These results are utilized in our computer program.

3. Specific Results

This section is divided into two parts; the first dealing with the algorithm for integration over rectangular regions and the second dealing with integration over elliptical areas in the bivariate case.

3.1 Integration Over Rectangular Areas

Suppose we desire to evaluate the integral

$$\theta = \int_{a_1}^{b_1}\int_{a_2}^{b_2}\cdots\int_{a_m}^{b_m} f(\underline{x}|\underline{\mu},\Sigma)dx. \tag{5}$$

Referring to formula (1) we define $g(\underline{x}) = 1/\prod_{i=1}^{m}(b_i-a_i)$ and evaluate the integral

$$\phi = \int_{a_1}^{b_1}\int_{a_2}^{a_1}\cdots\int_{a_m}^{b_m} f(\underline{x}|\underline{\mu},\Sigma)\frac{d\underline{x}}{\prod_{i=1}^{m}(b_i-a_i)}$$

and observe that we have

$$\theta = \phi \prod_{i=1}^{m}(b_i-a_i), \tag{6}$$

and the estimate

$$\hat{\theta} = \hat{\phi} \prod_{i=1}^{m}(b_i-a_i),$$

where

$$\hat{\phi} = \frac{1}{N}\Sigma\ f(\underline{x}_i|\underline{\mu},\Sigma)$$

when $\underline{x}_i$ is a random vector from the pdf

$$g(\underline{x}) = \begin{cases} \dfrac{1}{\prod_{i=1}^{m}(b_i-a_i)} & ;\ a_i \le x_i \le b_i,\ i=1,m. \\ 0 & \text{elsewhere} \end{cases}$$

This result is deceptively simple. The pdf $g(\underline{x})$ is the product of m independent uniform distributions and a random vector is generated using the equations $x_i = a_i + u_i(b_i - a_i)$, $i=1,m$, where u_i is a usual [0,1] random number.

In the program we obtain the $\Sigma = AA'$ factorization and calculate A^{-1} and $|A'|$. The algorithm proceeds to calculate $\underline{x}$, $\underline{y} = (A')^{-1}(\underline{x}-\underline{\mu})$ and $\underline{y}'\underline{y}$. Then $f(\underline{x}|\underline{\mu},\Sigma)$ follows directly.

This procedure converges slowly in most cases and an effort to increase the precision and speed of convergence was attempted by "removing the regular part" as generally described in section 2. Basically it consisted of expanding the expression $e^{-1/2(\underline{x}-\underline{\mu})\Sigma^{-1}(\underline{x}-\mu)}$ into a Taylor's series and defining $h(\underline{x})$ as the first three terms. The arithmetic is formidable and not germane to subsequent discussions. Consequently, it will not be presented in the text but can be obtained from the appended computer program.

The function $h(\underline{x})$ described above effectively decreases the standard error of the estimate when the integration ranges are small. To determine precisely where $h(\underline{x})$ fails, i.e., where $\mathrm{Var}(h(\underline{x})) > 2\,\mathrm{Cov}(f(\underline{x}))$ is a formidable task and not really necessary. To utilize this approach the computer must perform the calculations to evaluate (5) using the standard approach - the difference being the number of random number generations required to attain a specified level of the standard error of estimation (program input). At each 100 generations we check both errors and if either is below the specified level the computations terminate with the answer, standard error and an indication whether or not the regular part was removed being printed.

3.2 Elliptical Regions

This portion of the program has more direct applications to meteorological problems as the examples in the next section will show.

The general procedure is very similar to the case in section 3.1. Assuming we desire to integrate the bivariate normal pdf over an elliptical region E having area Πab where a and b are axis lengths, i.e., we desire to evaluate

$$\theta = \int_E\int f(\underline{x}|\underline{\mu},\Sigma)d\underline{x}. \tag{7}$$

Defining $g(\underline{x}) = 1/\Pi ab$ we evaluate

$$\phi = \int_E\int f(\underline{x}|\underline{\mu},\Sigma)d\underline{x}/\Pi ab$$

and, analogous to (6) above, obtain

$$\theta = \phi\ \Pi ab.$$

Here, as above, the problem is to generate a series of random vectors in the region being integrated over. The general procedure is to assume the ellipse is in standard position, i.e.,

$ES = \{(x,y)|\frac{x^2}{a^2} + \frac{y^2}{b^2} \leq 1\}$, obtain a random point in this ellipse and rotate and/or translate it to the area E. This random point in ES is obtained by generating two uniform [0,1] numbers u_1 and u_2. The equations $x = a(1-2u_i)$ and $y = b\sqrt{1-x^2/a^2}\ (1-2u_2)$ determine a random point on ES. Once translated and/or rotated to E the method in 3.1 applies except there is no provision for removing the regular part.

Program input permits considerable flexibility in the elliptical parameters required. Specifically, there are two options:

1) Inputs include x, y axis lengths, rotation angle and center after rotation or

2) Input 5 points on the locus of an ellipse (the program determines the ellipse and puts it in standard form).

4. Applications

This section presents some general numerical results and some examples. Specifically we present program outputs utilizing the data in Crutcher (1967). This also permits comparisons of our probabilities with those obtained by using the Cornell Aeronautical Laboratory tables (Groenewoud, et.al., 1967) on bivariate normal integrals over elliptical regions. We will, however, first present some computations over rectangular regions. The accuracy of our computations can be easily ascertained as they are merely products of independent standard normal integrals. Tables 1-4 present, respectively, the computations for the integrals

1. $\int_0^1 f(x|0,1)\, dx,$

2. $\int_{-1}^1 \int_{-1}^1 f(\underline{x}|\underline{0},I)\, d\underline{x},$

3. $\int_0^1\int_0^1\int_0^1 f(\underline{x}|\underline{0},I)\ d\underline{x}$

and

4. $\int_0^1\int_0^1\int_0^1\int_0^1 f(\underline{x}|\underline{0},I)\ d\underline{x}.$

The correct answers, using the normal probability tables in the C.R.C. Standard Mathematical Tables (Weast, 1968) are written in below the value calculated by the program.

TABLE 1

MULTIVARIATE NORMAL INTEGRAL

THE VECTOR OF MEANS IS
0.0000000

THE RANDOM NUMBER IS 5423

THE STANDARD ERROR IS LESS THAN 0.00100000

THE VARIANCE - COVARIANCE MATRIX S IS

1.00000	0.00000	0.00000	0.00000	0.00000
0.00000	0.00000	0.00000	0.00000	0.00000
0.00000	0.00000	0.00000	0.00000	0.00000
0.00000	0.00000	0.00000	0.00000	0.00000
0.00000	0.00000	0.00000	0.00000	0.00000

UPPER TRIANGULAR FACTORIZATION

1.00000

INVERSE OF A TRANSPOSE

1.00000

INVERSE OF S

1.00000

FOR RECTANGULAR REGION,DIMENSION = 1

THE LOWER INTEGRATION LIMITS ARE
0.0000000

THE UPPER INTEGRATION LIMITS ARE
1.0000002

THE VALUE IS 0.3412368 WITH A STANDARD ERROR OF 0.0001802
AND A CORRELATION OF 0.8611199

THE REGULAR PART HAS BEEN EXTRACTED

THE CORRECT VALUE IS .341345

TABLE 2

MULTIVARIATE NORMAL INTEGRAL

THE VECTOR OF MEANS IS
0.0000000 0.0000000

THE RANDOM NUMBER IS 8135

THE STANDARD ERROR IS LESS THAN 0.00100000

THE VARIANCE - COVARIANCE MATRIX S IS

1.00000	0.00000	0.00000	0.00000	0.00000
0.00000	1.00000	0.00000	0.00000	0.00000
0.00000	0.00000	0.00000	0.00000	0.00000
0.00000	0.00000	0.00000	0.00000	0.00000
0.00000	0.00000	0.00000	0.00000	0.00000

UPPER TRIANGULAR FACTORIZATION

1.00000	0.00000
0.00000	1.00000

INVERSE OF A TRANSPOSE

1.00000	0.00000
0.00000	1.00000

INVERSE OF S

1.00000	0.00000
0.00000	1.00000

FOR RECTANGULAR REGION,DIMENSION = 2

THE LOWER INTEGRATION LIMITS ARE
-1.0000002 -1.0000002

THE UPPER INTEGRATION LIMITS ARE
1.0000002 1.0000002

THE VALUE IS 0.4668157 WITH A STANDARD ERROR OF 0.0009669
AND A CORRELATION OF 0.8419297

THE REGULAR PART HAS BEEN EXTRACTED

THE CORRECT VALUE IS .46607

TABLE 3

MULTIVARIATE NORMAL INTEGRAL

THE VECTOR OF MEANS IS

0.0000000	0.0000000	0.0000000

THE RANDOM NUMBER IS 9631

THE STANDARD ERROR IS LESS THAN 0.00100000

THE VARIANCE - COVARIANCE MATRIX S IS

1.00000	0.00000	0.00000	0.00000	0.00000
0.00000	1.00000	0.00000	0.00000	0.00000
0.00000	0.00000	1.00000	0.00000	0.00000
0.00000	0.00000	0.00000	0.00000	0.00000
0.00000	0.00000	0.00000	0.00000	0.00000

UPPER TRIANGULAR FACTORIZATION

1.00000	0.00000	0.00000
0.00000	1.00000	0.00000
0.00000	0.00000	1.00000

INVERSE OF A TRANSPOSE

1.00000	0.00000	0.00000
0.00000	1.00000	0.00000
0.00000	0.00000	1.00000

INVERSE OF S

1.00000	0.00000	0.00000
0.00000	1.00000	0.00000
0.00000	0.00000	1.00000

FOR RECTANGULAR REGION,DIMENSION = 3

THE LOWER INTEGRATION LIMITS ARE

0.0000000	0.0000000	0.0000000

THE UPPER INTEGRATION LIMITS ARE

1.0000002	1.0000002	1.0000002

THE VALUE IS 0.0397351 WITH A STANDARD ERROR OF 0.0002741
AND A CORRELATION OF 0.8054759

THE REGULAR PART HAS BEEN EXTRACTED

THE CORRECT VALUE IS .039772

TABLE 4

MULTIVARIATE NORMAL INTEGRAL

THE VECTOR OF MEANS IS

0.0000000	0.0000000	0.0000000	0.0000000

THE RANDOM NUMBER IS 7317

THE STANDARD ERROR IS LESS THAN 0.00100000

THE VARIANCE - COVARIANCE MATRIX S IS

1.00000	0.00000	0.00000	0.00000	0.00000
0.00000	1.00000	0.00000	0.00000	0.00000
0.00000	0.00000	1.00000	0.00000	0.00000
0.00000	0.00000	0.00000	1.00000	0.00000
0.00000	0.00000	0.00000	0.00000	0.00000

UPPER TRIANGULAR FACTORIZATION

1.00000	0.00000	0.00000	0.00000
0.00000	1.00000	0.00000	0.00000
0.00000	0.00000	1.00000	0.00000
0.00000	0.00000	0.00000	1.00000

INVERSE OF A TRANSPOSE

1.00000	0.00000	0.00000	0.00000
0.00000	1.00000	0.00000	0.00000
0.00000	0.00000	1.00000	0.00000
0.00000	0.00000	0.00000	1.00000

INVERSE OF S

1.00000	0.00000	0.00000	0.00000
0.00000	1.00000	0.00000	0.00000
0.00000	0.00000	1.00000	0.00000
0.00000	0.00000	0.00000	1.00000

FOR RECTANGULAR REGION, DIMENSION = 4

THE LOWER INTEGRATION LIMITS ARE

0.0000000	0.0000000	0.0000000	0.0000000

THE UPPER INTEGRATION LIMITS ARE

1.0000002	1.0000002	1.0000002	1.0000002

THE VALUE IS 0.0135361 WITH A STANDARD ERROR OF 0.0001962
AND A CORRELATION OF 0.8278033

THE CORRECT VALUE IS .013576

The output is self-explanatory except for the standard error value. The value .001 is used in each case. This quantity is the standard error of the estimated integral value. In Table 1 the estimated value is .341236 with a standard error of .00018 and, consequently, a 95% confidence interval on the "true" value of the integral is .341236 $\pm$ .00035 which is observed to contain the true value in this case.

The above paragraph demonstrates the basic statistical approach of the Monte Carlo technique. We are estimating a parameter in the strictest sense and this estimate has the same validity of an estimate for any quantity when many samples (likely several thousand in most cases) are involved. Here, as in most applications, the well-known mechanism of increasing the precision by increasing the sample size and/or decreasing the population variance (as the "regular part" does in some instances) is working. The confidence placed in these numbers should be no more or no less than the confidence placed in the results of any carefully designed, controlled and extensive sampling scheme.

The next three tables (5-7) present the calculated probabilities for examples H, I and J (pp. 25-31) in Crutcher (1967). Briefly, the examples are presented below:

Example H: 24-Hour Displacement of Asiatic East Coast Cyclones Originating East of 130° East Longitude.

Bivariate Normal Parameters (Units in latitude)

$\bar{x} = 8.9$ $\quad S_x = 3.10$ $\quad r_{xy} = .292$

$\bar{y} = 4.6$ $\quad S_y = 3.70$

The problem is to evaluate the probability of observing a cyclone after 24-hours in a circle centered due east of the original position at a distance of 8 degrees of latitude with a radius of 2 degrees.

Example I: Upper Wind Velocities for January at 6 km above Greensboro, North Carolina.

Bivariate Normal Parameters (Units are meters/sec. with winds from south and west positive)

$\bar{x} = 24.04$ $\quad S_x = 13.12$ $\quad r_{xy} = .174$

$\bar{y} = 1.22$ $\quad S_y = 12.65$

The problem is to evaluate the probability of observing winds at 6 km equal to or less than 5 meters/sec.

Example J: 36-Hour Displacement of Tropical Storms.

Bivariate Normal Parameters (Units are degrees of latitude)

$\bar{x} = -2.9$ $\quad S_x = 3.87$ $\quad r_{xy} = .581$

$\bar{y} = 3.9$ $\quad S_y = 2.30$

> The problem is to evaluate the probability of observing a storm within 2 degrees latitude of a point (-5.0, 5.0) from the storm center 36 hours previously.

On each table the probability obtained by using the Cornell Tables and the exact calculated probability, furnished by Crutcher (1967), are given.

TABLE 5

MULTIVARIATE NORMAL INTEGRAL

THE VECTOR OF MEANS IS
8.9000015 4.6000003

THE RANDOM NUMBER IS 3527

THE STANDARD ERROR IS LESS THAN 0.00100000

THE VARIANCE - COVARIANCE MATRIX S IS

9.61000	3.34923	0.00000	0.00000	0.00000
3.34923	13.68999	0.00000	0.00000	0.00000
0.00000	0.00000	0.00000	0.00000	0.00000
0.00000	0.00000	0.00000	0.00000	0.00000
0.00000	0.00000	0.00000	0.00000	0.00000

UPPER TRIANGULAR FACTORIZATION

3.09999	1.08039
0.00000	3.53874

INVERSE OF A TRANSPOSE

0.32258	0.00000
-0.09848	0.28258

INVERSE OF S

0.11375	-0.02783
-0.02783	0.07985

FOR BIVARIATE ELLIPTICAL REGION

THE ROTATION ANGLE IS 0.00000
THE X AND Y AXIS LENGTHS ARE 2.0000 2.0000
THE CENTER AFTER ROTATION IS 8.0000 0.0000

THE VALUE IS 0.0783936 WITH A STANDARD ERROR OF 0.0009339

VALUE USING CORNELL TABLES IS .07733

EXACT CALCULATED VALUE IS .07713

TABLE 6

MULTIVARIATE NORMAL INTEGRAL

THE VECTOR OF MEANS IS
 24.0400009   1.2200000

THE RANDOM NUMBER IS 1179

THE STANDARD ERROR IS LESS THAN 0.00100000

THE VARIANCE - COVARIANCE MATRIX S IS

172.13439	28.87842	0.00000	0.00000	0.00000
28.87842	160.02249	0.00000	0.00000	0.00000
0.00000	0.00000	0.00000	0.00000	0.00000
0.00000	0.00000	0.00000	0.00000	0.00000
0.00000	0.00000	0.00000	0.00000	0.00000

UPPER TRIANGULAR FACTORIZATION

13.11999	2.20110
0.00000	12.45703

INVERSE OF A TRANSPOSE

0.07621	0.00000
-0.01346	0.08027

INVERSE OF S

0.00599	-0.00108
-0.00108	0.00644

FOR BIVARIATE ELLIPTICAL REGION

THE ROTATION ANGLE IS 0.00000
THE X AND Y AXIS LENGTHS ARE 5.0000 5.0000
THE CENTER AFTER ROTATION IS 0.0000 0.0000

THE VALUE IS 0.0146952 WITH A STANDARD ERROR OF 0.0006107

VALUE USING CORNELL TABLES IS .01474

EXACT CALCULATED VALUE IS .01431

TABLE 7

MULTIVARIATE NORMAL INTEGRAL

THE VECTOR OF MEANS IS
-2.9000001 3.9000001

THE RANDOM NUMBER IS 2399

THE STANDARD ERROR IS LESS THAN 0.00100000

THE VARIANCE - COVARIANCE MATRIX S IS

14.97690	5.17148	0.00000	0.00000	0.00000
5.17148	5.28999	0.00000	0.00000	0.00000
0.00000	0.00000	0.00000	0.00000	0.00000
0.00000	0.00000	0.00000	0.00000	0.00000
0.00000	0.00000	0.00000	0.00000	0.00000

UPPER TRIANGULAR FACTORIZATION

3.86999	1.33629
0.00000	1.87197

INVERSE OF A TRANSPOSE

0.25839	0.00000
-0.18445	0.53419

INVERSE OF S

0.10079	-0.09853
-0.09853	0.28536

FOR BIVARIATE ELLIPTICAL REGION

THE ROTATION ANGLE IS 0.00000
THE X AND Y AXIS LENGTHS ARE 2.0000 2.0000
THE CENTER AFTER ROTATION IS -5.0000 5.0000

THE VALUE IS 0.1445387 WITH A STANDARD ERROR OF 0.0009946

VALUE USING CORNELL TABLES IS .14363

EXACT CALCULATED VALUE IS .14546

The Monte Carlo technique is a technique of considerable value in evaluating complicated integrals. From the two cases presented here the reader can readily grasp the importance of a proper selection of the pdf $g(\underline{x})$. If the region being integrated over can be described mathematically the process is readily applicable - the form of the integrand presents no problem if it can be evaluated accurately.

References

Crutcher, H. L. (1967). Bivariate Normal Offset Circle Probability Tables, Vol. III. Cornell Aeronautical Laboratory, Inc. Buffalo, N. Y.

Groenewoud, C., Hoaglin, D. C. and Vitalis, J. A. (1967). Bivariate Normal Offset Circles Probability Tables, Vols. I and II. Cornell Aeronautical Laboratory, Inc. Buffalo, N. Y.

Newman, T. G. and Odell, P. L. (1971). The Generation of Random Variates. Hafner, New York.

Shreider, Y. A. (1964). Monte Carlo Methods. Elsevier, New York.

Weast, R. C. (Ed.) (1968). Standard Mathematical Tables, 16th ed. Chemical Rubber Company, Cleveland, Ohio.

Appendix

This appendix provides a detailed description of the required card inputs to utilize the programs, along with appropriate JCL for the computer being utilized. The programs are written in Fortran IV.

The changes necessary to adapt the program to a specific computer will be the modification to utilize a random number generator other than the IBM SSP routine called RANDU and assigning the proper logical unit numbers to reader and printer. The former can be accomplished by changing Format statement number 90 to conform to prescribed starting integer length (presently the variable designation is Ix read in I4 format) and the latter can be accomplished by changing the R = 2 and P = 3 cards at the beginning of the program to designate the reader and printer logical numbers.

Description - Card Input

	COL.	
CARD 1	1	DIMENSION OF INTEGRAL
	2	0 = RECTANGULAR INTEGRATION, 1 = ELLIPSE
	3	0 = INPUT IS X & Y AXIS LENGTHS, ROTATION ANGLE & CENTER AFTER ROTATION
		1 = INPUT IS 5 POINTS ON ELLIPSE
	4-7	4 DIGIT RANDOM NUMBER (LAST DIGIT ODD)
	8-17	MAXIMUM STD ERROR OF THE MEAN PERMITTED (F10.6 FORMAT)

CARD 2 VECTOR OF MEANS (MAXIMUM OF 5) 5F8.3 FORMAT

		FIELD	COLS	
CARDS 3 to 3+N-1	1st CARD	1	1-8	STD DEVIATION OF 1st VARIABLE
		2	9-16	CORRELATION OF 1st & 2nd VARIABLE
		3	17-24	CORRELATION OF 1st & 3rd VARIABLE
		4	25-32	CORRELATION OF 1st & 4th VARIABLE
		5	33-40	CORRELATION OF 1st & 5th VARIABLE
	2nd CARD	1	1-8	STD DEVIATION OF 2nd VARIABLE
		2	9-16	CORRELATION OF 2nd & 3rd VARIABLES
		3	17-24	CORRELATION OF 2nd & 4th VARIABLES
		4	25-32	CORRELATION OF 2nd & 5th VARIABLES
	3rd CARD	etc.		

NOTE: THERE WILL BE AS MANY CARDS AS THE DIMENSION OF THE INTEGRAL - EACH CARD CONTAINING THE STD DEVIATION OF A VARIABLE AND CORRELATIONS WITH SUCCEEDING VARIABLES WITH 1st CARD DEVOTED TO VARIABLE 1, 2nd CARD TO VARIABLE 2, ETC.

IF CARD 1, COL 2 = 0

THE NEXT TWO CARDS CONTAIN THE LIMITS OF INTEGRATION

THE FIRST CONTAINING THE LOWER LIMITS, THE SECOND CONTAINING THE UPPER LIMITS. BOTH CARDS HAVE 5F8.3 FORMATS.

1st CARD	FIELD 1	COLS 1-8	LOWER INTEGRATION LIMIT ON VARIABLE 1
	2	9-16	LOWER INTEGRATION LIMIT ON VARIABLE 2
	3	17-24	LOWER INTEGRATION LIMIT ON VARIABLE 3
	4	25-32	LOWER INTEGRATION LIMIT ON VARIABLE 4
	5	33-40	LOWER INTEGRATION LIMIT ON VARIABLE 5
2nd CARD	FIELD 1	COLS 1-8	UPPER INTEGRATION LIMIT ON VARIABLE 1
	2	9-16	UPPER INTEGRATION LIMIT ON VARIABLE 2
	3	17-24	UPPER INTEGRATION LIMIT ON VARIABLE 3
	4	25-32	UPPER INTEGRATION LIMIT ON VARIABLE 4
	5	33-40	UPPER INTEGRATION LIMIT ON VARIABLE 5

IF CARD 1, COL 2 = 1

THE NEXT CARD CONTAINS ELLIPSE INFORMATION

IF CARD 1, COL 3 = 0

FIELD 1	COLS 1-10	ROTATION ANGLE IN RADIANS F10.3 FORMAT
2	11-20	X AXIS LENGTH (in regular position) F10.3 FORMAT
3	21-30	Y AXIS LENGTH (in regular position) F10.3 FORMAT
4	31-40	X COORDINATE OF CENTER F10.3 FORMAT
5	41-50	Y COORDINATE OF CENTER F10.3 FORMAT

IF CARD 1, COL 3 = 1

FIELD 1	COLS 1-6	X COORDINATE OF 1st POINT
2	7-12	Y COORDINATE OF 1st POINT F6.2 FORMAT
3	13-18	X COORDINATE OF 2nd POINT F6.2 FORMAT
4	19-24	Y COORDINATE OF 2nd POINT F6.2 FORMAT
5	25-30	X COORDINATE OF 3rd POINT F6.2 FORMAT

6	31-36	Y COORDINATE OF 3rd POINT F6.2 FORMAT
7	37-42	X COORDINATE OF 4th POINT F6.2 FORMAT
8	43-48	Y COORDINATE OF 4th POINT F6.2 FORMAT
9	49-54	X COORDINATE OF 5th POINT F6.2 FORMAT
10	55-60	Y COORDINATE OF 5th POINT F6.2 FORMAT

```
      DIMENSION X(5),S(5,5),A(5),B(5),E(5,5),H(5,5),Y(5),SD(5),RHO(5,5),
     1KN(5),KI(5),F(5,5),U(5),AXIS(2),PTS(2),AX(2),AY(2),XBR(5)
C
   90 FORMAT(3I1,I4,F10.6)
   91 FORMAT(5F8.3)
   93 FORMAT('0THE VARIANCE - COVARIANCE MATRIX S IS'/)
   94 FORMAT(/,' WRONG INPUT PARAMETER')
   95 FORMAT(/,' LOSS OF SIGNIFICANCE')
   96 FORMAT(//,' UPPER TRIANGULAR FACTORIZATION ',/)
   97 FORMAT(1X,5(2X,F10.5))
   98 FORMAT(//,' INVERSE OF A TRANSPOSE',/)
   99 FORMAT(//,' INVERSE OF S',/)
  100 FORMAT('0THE VALUE IS',F15.10,' WITH A STANDARD ERROR OF',F15.10,/
     1/,' THE VALUE IS CALCULATED WITHOUT REMOVING THE REGULAR PART')
  104 FORMAT('0THE VALUE IS',F15.10,' WITH A STANDARD ERROR OF',F15.10,'
     1 AND A CORRELATION OF',F15.10,//' THE REGULAR PART IS POSITIVELY C
     2ORRELATED WITH THE INTEGRAL AND IS THUS EXTRACTED')
C
      EPS=.000001
    1 CONTINUE
      DO 112 I=1,5
      DO 112 J=1,5
  112 S(I,J)=0.
      WRITE(3,8)
    8 FORMAT('1',20X,29HMULTIVARIATE NORMAL INTEGRAL /)
      NT= 100
      READ(2,90) N,KODE,ICODE,IX,ERROR
      READ(2,91) (U(I),I=1,N)
      DO 2 L=1,N
      LL=L+1
    2 READ(2,91) SD(L),(RHO(L,K),K=LL,N)
      WRITE(3,105) (U(I),I=1,N)
  105 FORMAT('0THE VECTOR OF MEANS IS'/,1X,5(F12.7,3X))
      WRITE(3,107) IX,ERROR
  107 FORMAT(//,' THE RANDOM NUMBER IS',I6// ' THE STANDARD ERROR IS LES
     1S THAN',F12.8)
      DO 3 I=1,N
      LL=I+1
      S(I,I)=SD(I)**2
      DO 3 L=LL,N
      S(I,L)=RHO(I,L)*SD(I)*SD(L)
    3 S(L,I)=S(I,L)
      WRITE(3,93)
      DO 4 I=1,5
    4 WRITE(3,97) (S(I,J),J=1,5)
C     FACTOR MATRIX S
      DO 10 I=1,N
      DO 10 J=1,N
      E(I,J)=S(I,J)
   10 H(I,J)=S(I,J)
      CALL ARRAY(2,N,N,5,5,E,E)
      CALL MSTR(E,F,N,0,1)
      CALL MFSD(F,N,EPS,IER)
      CALL MSTR(F,E,N,1,0)
      CALL ARRAY(1,N,N,5,5,E,E)
      IF(N-1) 11,13,11
   11 NN=N-1
      DO 12 J=1,NN
```

```
      IFH=J+1
      DO 12 I=IFH,N
12    E(I,J)=0.0
13    IF(IER) 14,16,15
14    WRITE(3,94)
      GO TO 16
15    WRITE(3,95)
16    WRITE(3,96)
      DO 17 I=1,N
17    WRITE(3,97) (E(I,J),J=1,N)
C
C     INVERT A
C
      DO 18 I=1,N
      DO 18 J=1,N
18    F(J,I)=E(I,J)
      CALL ARRAY(2,N,N,5,5,F,F)
      CALL MINV(F,N,DU,KN,KI)
      CALL ARRAY(1,N,N,5,5,F,F)
      IF(IER) 19,21,20
19    WRITE(3,94)
      GO TO 21
20    WRITE(3,95)
21    WRITE(3,98)
      DO 22 I=1,N
22    WRITE(3,97) (F(I,J),J=1,N)
      CALL ARRAY(2,N,N,5,5,H,H)
      CALL MSTR(H,S,N,0,1)
      CALL SINV(S,N,EPS,IER)
      CALL MSTR(S,H,N,1,0)
      CALL ARRAY(1,N,N,5,5,H,H)
      IF(IER) 23,25,24
23    WRITE(3,94)
      GO TO 25
24    WRITE(3,95)
25    WRITE(3,99)
      DO 26 I=1,N
26    WRITE(3,97) (H(I,J),J=1,N)
C
C     CALCULATE CONSTANT C
C
      PROD=1.
      DO 30 I=1,N
30    PROD=PROD*E(I,I)
      PPROD=PROD**2
      C=((6.283185307)**N*PPROD)**(-.5)
      IF(N-2) 32,31,32
C
C     IF KODE = 1, THIS IS AN ELLIPSE
C
31    IF(KODE) 32,32,200
C
C     CALCULATE QUANTITY Q
C
   32 WRITE(3,111)N
  111 FORMAT(21X,'FOR RECTANGULAR AREA,DIMENSION =',I2/)
```

```
      P1=1.
      READ(2,91) (A(I),I=1,N)
      READ(2,91) (B(I),I=1,N)
      WRITE(3,106) (A(I),I=1,N)
  106 FORMAT('0THE LOWER INTEGRATION LIMITS ARE'/,1X,5(F12.7,3X))
      WRITE(3,108) (B(I),I=1,N)
  108 FORMAT('0THE UPPER INTEGRATION LIMITS ARE'/,1X,5(F12.7,3X))
      DO 33 I=1,N
33    P1=P1*(B(I)-A(I))
      TS1=0
      SUM1=0
      DO 36 I=1,N
      DIF1=(B(I)-U(I))**3-(A(I)-U(I))**3
      TE2=(B(I)-U(I))**5-(A(I)-U(I))**5
      P2=1.
      DO 35 J=1,N
      IF(J-I)34,35,34
   34 P2=P2*(B(J)-A(J))
   35 CONTINUE
      SUM1=SUM1+(H(I,I)*DIF1*P2)
   36 TS1=TS1+(H(I,I)**2*TE2*P2)
      SUM1=(-.5)*SUM1/3.
      TE1=.025*TS1
      SUM2=0.0
      IF(N-1) 42,42,37
37    DO 41 J=1,NN
      II=J+1
      DO 41 I=II,N
      DIF2=(B(I)-U(I))**2-(A(I)-U(I))**2
      DIF3=(B(J)-U(J))**2-(A(J)-U(J))**2
      P3=1.
      DO 40 K=1,N
      IF(K-I) 38,40,38
38    IF(K-J) 39,40,39
39    P3=P3*(B(K)-A(K))
40    CONTINUE
      SUM2=SUM2+(H(I,J)*DIF2*DIF3*P3)
41    CONTINUE
   42 Q=C*(P1+SUM1-SUM2/4.+TE1)
      TS2=0
      TS3=0
      TS4=0
      IF(N-1)65,65,66
   66 DO 67 I=1,N
      DO 67 J=1,N
      IF(J-I)69,67,69
   69 DS1=((B(I)-U(I))**4-(A(I)-U(I))**4)*((B(J)-U(J))**2-(A(J)-U(J))**2
     1)
      DS2=((B(I)-U(I))**3-(A(I)-U(I))**3)*((B(J)-U(J))**3-(A(J)-U(J))**3
     1)
      P2=1
      DO 70 K=1,N
      IF (K-I)71,70,71
   71 IF (K-J)72,70,72
   72 P2=P2*(B(K)-A(K))
   70 CONTINUE
```

```
      TS2=TS2+H(I,I)*H(I,J)*DS1*P2
      TS3=TS3+(H(I,I)*H(J,J)+2.*H(I,J)**2)*DS2*P2
   67 CONTINUE
      TS2=TS2/16.
      TS3=TS3/72.
      IF(N-2)65,65,73
   73 DO 74 I=1,N
      DO 74 J=1,N
      IF(J-I)75,74,75
   75 DO 74 K=1,N
      IF(K-I)76,74,76
   76 IF(K-J)77,74,77
   77 DS1=(B(I)-U(I))**3-(A(I)-U(I))**3
      DS1=DS1*((B(J)-U(J))**2-(A(J)-U(J))**2)
      DS1=DS1*((B(K)-U(K))**2-(A(K)-U(K))**2)
      P2=1
      DO 78 L=1,N
      IF(L-I)79,78,79
   79 IF(L-J)80,78,80
   80 IF(L-K)81,78,81
   81 P2=P2*(B(L)-A(L))
   78 CONTINUE
      TS4=TS4+(2.*H(I,I)*H(J,K)+4.*H(I,J)*H(I,K))*DS1*P2
   74 CONTINUE
      TS4=TS4/96.
   65 Q=Q+C*(TS2+TS3+TS4)
C
C     CALCULATE C'
C
      CP=P1*C
C
C     START RANDOM PROCESS
C     CALCULATE X (RANDOM VECTOR)
C
      FSUM=0.0
      FSQ=0.
      FGSUM=0.
      FGSQ=0.
      COFG=0.
   50 DO 59 NNN=1,100
      DO 51 I=1,N
      CALL RANDU(IX,IY,YFL)
      IX=IY
   51 X(I)=YFL
      DO 52 I=1,N
   52 Y(I)=A(I)+X(I)*(B(I)-A(I))
C
C     CALCULATE F AND G
C
      DO 53 I=1,N
      XBR(I)=0.
      DO 53 J=1,N
   53 XBR(I)=XBR(I)+F(I,J)*(Y(J)-U(J))
      T=1.
      DO 54 I=1,N
   54 T=T*EXP(-.5*XBR(I)**2)
```

```
      FF=T*CP
      SUM3=0.
      TS1=0
      TS2=0
      TS3=0
      DO 55 I=1,N
      SUM3=SUM3+H(I,I)*(Y(I)-U(I))**2
   55 TS1=TS1+H(I,I)**2*(Y(I)-U(I))**4
      SUM4=0.
      IF(N-1) 58,58,56
56    DO 57 J=1,NN
      IM=J+1
      DO 57 I=IM,N
57    SUM4=SUM4+H(I,J)*((Y(J)-U(J))*(Y(I)-U(I)))
      DO 82 I=1,N
      DO 82 J=1,N
      IF(J-I)83,82,83
   83 DI1=(Y(I)-U(I))**3*(Y(J)-U(J))
      DI2=(Y(I)-U(I))**2*(Y(J)-U(J))**2
      TS2=TS2+(4.*H(I,I)*H(I,J)*DI1+(H(I,I)*H(J,J)+2.*H(I,J)**2)*DI2)
   82 CONTINUE
      IF(N-2)84,84,85
   85 DO 86 I=1,N
      DO 86 J=1,N
      IF (J-I)87,86,87
   87 DO 86 K=1,N
      IF (K-I)88,86,88
   88 IF(K-J)89,86,89
   89 DI1=(Y(I)-U(I))**2*(Y(J)-U(J))*(Y(K)-U(K))
      TS3=TS3+(2.*H(I,I)*H(J,K)+4.*H(I,J)*H(I,K))*DI1
   86 CONTINUE
   84 CONTINUE
   58 G=CP*(1.-.5*(SUM3+2.*SUM4)+.125*(TS1+TS2+TS3))
      FSUM=FSUM+FF
      FSQ=FSQ+(FF**2)
      FGSUM=FGSUM+(FF-G)
      FGSQ=FGSQ+((FF-G)**2)
      COFG=COFG+FF*(FF-G)
52    CONTINUE
      FM=FSUM/NT
      VAR1=(FSQ-(FSUM**2/NT))/(NT-1)
      COF=(COFG-FSUM*FGSUM/NT)/(NT-1.)
      SF=SQRT(VAR1/NT)
      FGM=FGSUM/NT
      VAR2=(FGSQ-(FGSUM**2)/NT)/(NT-1)
      SFG=SQRT(VAR2/NT)
      FG=FGM+Q
      IF (SFG-ERROR)63,63,60
   60 IF(SF-ERROR)62,62,61
   61 NT=NT+100
      GO TO 50
   62 WRITE(3,100)FM,SF
      GO TO 1
   63 COF=COF/SQRT(VAR1*VAR2)
      WRITE(3,104)FG,SFG,COF
      GO TO 1
```

```
  200 CONTINUE
  103  FORMAT('0THE VALUE IS',F15.10,' WITH A STANDARD ERROR OF',F15.10)
  101 FORMAT(5F10.3)
  102 FORMAT('0THE ROTATION ANGLE IS',F10.5,/,' THE X AND Y AXIS LENGTHS
     1 ARE',2(2X,F10.4),/,' THE CENTER AFTER ROTATION IS',2(2X,F10.4))
      WRITE(3,110)
  110 FORMAT(21X,'FOR ELLIPTICAL REGION'/)
      IF (ICODE) 211,211,212
  212 CALL CAL(THETA,AXIS,PTS)
      GO TO 213
  211 READ(2,101) THETA,AXIS,PTS
  213 WRITE(3,102) THETA,AXIS,PTS
      C=3.141592653*C*AXIS(1)*AXIS(2)
      SINT=SIN(THETA)
      COST=COS(THETA)
      FSUM=0.
      FSQ=0.
  201 DO 204 NNN = 1,100
      CALL RANDU(IX,IY,YFL)
      IX=IY
      AX(1)=-AXIS(1)+2.*YFL*AXIS(1)
      CALL RANDU(IX,IY,YFL)
      IX=IY
      DISC=1-(AX(1)**2)/(AXIS(1)**2)
      AX(2)=-AXIS(2)*SQRT(DISC)+2.*YFL*AXIS(2)*SQRT(DISC)
      AY(1)=AX(1)*COST-AX(2)*SINT+PTS(1)
      AY(2)=AX(1)*SINT+AX(2)*COST+PTS(2)
      DO 202 I=1,N
      XBR(I)=0.0
      DO 202 J=1,N
  202 XBR(I)=XBR(I)+F(I,J)*(AY(J)-U(J))
      T=1.
      DO 203 I=1,N
  203 T=T*EXP(-.5*XBR(I)**2)
      FF=T*C
      FSUM=FSUM+FF
      FSQ=FSQ+(FF**2)
  204 CONTINUE
      FM=(FSUM)/NT
      VAR1=(FSQ-(FSUM**2/NT))/(NT-1)
      SF=SQRT(VAR1/NT)
  210 IF(SF-ERROR) 206,205,205
  205 NT=NT+100
      GO TO 201
  206 WRITE(3,103)FM,SF
  300 GO TO 1
56789 STOP
      END
```

```
      SUBROUTINE LOC(I,J,IR,N,M,MS)
      IX=I
      JX=J
      IF(MS-1) 10,20,30
10    IRX=N*(JX-1)+IX
      GO TO 36
20    IF(IX-JX) 22,24,24
22    IRX=IX-(JX*JX-JX)/2.
      GO TO 36
24    IRX=JX+(IX*IX-IX)/2
      GO TO 36
30    IRX=0
      IF(IX-JX) 36,32,36
32    IRX=IX
36    IR=IRX
      RETURN
      END
```

```
      SUBROUTINE MSTR(A,R,N,MSA,MSR)
      DIMENSION A(1),R(1)
      DO 20 I=1,N
      DO 20 J=1,N
      IF(MSR) 5,10,5
    5 IF(I-J) 10,10,20
   10 CALL LOC(I,J,IR,N,N,MSR)
      IF(IR) 20,20,15
   15 R(IR)=0.0
      CALL LOC(I,J,IA,N,N,MSA)
      IF(IA) 20,20,18
   18 R(IR)=A(IA)
   20 CONTINUE
      RETURN
      END
```

```
      SUBROUTINE MINV(A,N,D,L,M)
      DIMENSION A(1),L(1),M(1)
      D=1.0
      NK=-N
      DO 80 K=1,N
      NK=NK+N
      L(K)=K
      M(K)=K
      KK=NK+K
      BIGA=A(KK)
      DO 20 J=K,N
      IZ=N*(J-1)
      DO 20 I=K,N
      IJ=IZ+I
   10 IF(ABS(BIGA)-ABS(A(IJ)))15,20,20
   15 BIGA=A(IJ)
      L(K)=I
      M(K)=J
   20 CONTINUE
      J=L(K)
      IF(J-K)35,35,25
   25 KI=K-N
      DO 30 I=1,N
      KI=KI+N
      HOLD=-A(KI)
      JI=KI-K+J
      A(KI)=A(JI)
   30 A(JI)=HOLD
   35 I=M(K)
      IF(I-K)45,45,38
   38 JP=N*(I-1)
      DO 40 J=1,N
      JK=NK+J
      JI=JP+J
      HOLD=-A(JK)
      A(JK)=A(JI)
   40 A(JI)=HOLD
   45 IF(BIGA)48,46,48
   46 D=0.0
      RETURN
   48 DO 55 I=1,N
      IF(I-K)50,55,50
   50 IK=NK+I
      A(IK)=A(IK)/(-BIGA)
   55 CONTINUE
      DO 65 I=1,N
      IK=NK+I
      HOLD=A(IK)
      IJ=I-N
      DO 65 J=1,N
      IJ=IJ+N
      IF(I-K)60,65,60
   60 IF(J-K)62,65,62
   62 KJ=IJ-I+K
      A(IJ)=HOLD*A(KJ)+A(IJ)
   65 CONTINUE
      KJ=K-N
      DO 75 J=1,N
```

```
      KJ=KJ+N
      IF(J-K)70,75,70
   70 A(KJ)=A(KJ)/BIGA
   75 CONTINUE
      D=D*BIGA
      A(KK)=1.0/BIGA
   80 CONTINUE
      K=N
  100 K=K-1
      IF(K)150,150,105
  105 I=L(K)
      IF(I-K)120,120,108
  108 JQ=N*(K-1)
      JR=N*(I-1)
      DO 110 J=1,N
      JK=JQ+J
      HOLD=A(JK)
      JI=JR+J
      A(JK)=-A(JI)
  110 A(JI)=HOLD
  120 J=M(K)
      IF(J-K)100,100,125
  125 KI=K-N
      DO 130 I=1,N
      KI=KI+N
      HOLD=A(KI)
      JI=KI-K+J
      A(KI)=-A(JI)
  130 A(JI)=HOLD
      GO TO 100
  150 RETURN
      END
```

```
      SUBROUTINE ARRAY(MODE,I,J,N,M,S,D)
      DIMENSION S(1),D(1)
      NI=N-I
C     TEST TYPE OF CONVERSION
      IF(MODE-1) 100,100,120
C     CONVERT FROM SINGLE TO DOUBLE DIMENSION
100   IJ=I*J+1
      NM=N*J+1
      DO 110 K=1,J
      NM=NM-NI
      DO 110 L=1,I
      IJ=IJ-1
      NM=NM-1
110   D(NM)=S(IJ)
      GO TO 140
C     CONVERT FROM DOUBLE TO SINGLE DIMENSION
120   IJ=0
      NM=0
      DO 130 K=1,J
      DO 125 L=1,I
      IJ=IJ+1
      NM=NM+1
125   S(IJ)=D(NM)
130   NM=NM+NI
140   RETURN
      END
```

```
      SUBROUTINE MFSD(A,N,EPS,IER)
      DIMENSION A(1)
C     TEST ON WRONG INPUT PARAMETER N
      IF(N-1) 12,1,1
1     IER=0
C     INITIALIZE DIAGONAL LOOP
      KPIV=0
      DO 11 K=1,N
      KPIV=KPIV+K
      IND=KPIV
      LEND=K-1
C     CALCULATE TOLERANCE
      TOL=ABS(EPS*A(KPIV))
C     START FACTORIZATION LOOP OVER K-TH ROW
      DO 11 I=K,N
      DSUM=0.
      IF(LEND) 2,4,2
C     START INNER LOOP
2     DO 3 L=1,LEND
      LANF=KPIV-L
      LIND=IND-L
3     DSUM=DSUM+A(LANF)*A(LIND)
C     END OF INNER LOOP
C     TRANSFORM ELEMENT A(IND)
4     DSUM=A(IND)-DSUM
      IF(I-K) 10,5,10
C     TEST FOR NEGATIVE PIVOT ELEMENT AND FOR LOSS OF SIGNIFICANCE
5     IF(DSUM-TOL) 6,6,9
6     IF(DSUM) 12,12,7
7     IF(IER) 8,8,9
8     IER=K-1
C     COMPUTE PIVOT ELEMENT
9     DPIV=SQRT(DSUM)
      A(KPIV)=DPIV
      DPIV=1./DPIV
      GO TO 11
C     CALCULATE TERMS IN ROW
10    A(IND)=DSUM*DPIV
11    IND=IND+I
C     END OF DIAGONAL LOOP
      RETURN
12    IER=-1
      RETURN
      END
```

```
      SUBROUTINE SINV(A,N,EPS,IER)
      DIMENSION A(1)
      CALL MFSD(A,N,EPS,IER)
      IF(IER) 9,1,1
    1 IPIV=N*(N+1)/2
      IND=IPIV
      DO 6 I=1,N
      DIN=1.0/A(IPIV)
      A(IPIV)=DIN
      MIN=N
      KEND=I-1
      LANF=N-KEND
      IF(KEND) 5,5,2
    2 J=IND
      DO 4 K=1,KEND
      WORK=0.0
      MIN=MIN-1
      LHOR=IPIV
      LVER=J
      DO 3 L=LANF,MIN
      LVER=LVER+1
      LHOR=LHOR+L
    3 WORK=WORK+A(LVER)*A(LHOR)
      A(J)=-WORK*DIN
    4 J=J-MIN
    5 IPIV=IPIV-MIN
    6 IND=IND-1
      DO 8 I=1,N
      IPIV=IPIV+I
      J=IPIV
      DO 8 K=I,N
      WORK=0.0
      LHOR=J
      DO 7 L=K,N
      LVER=LHOR+K-I
      WORK=WORK+A(LHOR)*A(LVER)
    7 LHOR=LHOR+L
      A(J)=WORK
    8 J=J+K
    9 RETURN
      END
```

```
      SUBROUTINE CAL(THETA,PTS,AXIS)
      DIMENSION A(5,2),S(5,5),BP(6),B(6),PTS(2),AXIS(2),KN(5),KI(5)
   10 READ(2,100) ((A(I,J),J=1,2),I=1,5)
  100 FORMAT(10F6.2)
      WRITE(3,101) ((A(I,J),J=1,2),I=1,5)
  101 FORMAT(//,' THE POINTS ARE',/,5(2X,'(',2F7.2,' )',))
      DO 1 I=1,5
      S(I,1)=A(I,1)**2.
      S(I,2)=A(I,1)*A(I,2)
      S(I,3)=A(I,2)**2
      S(I,4)=A(I,1)
    1 S(I,5)=A(I,2)
      CALL MINV(S,5,DU,KN,KI)
      DO 2 I=1,5
      SUM=0.0
      DO 3 J=1,5
    3 SUM=SUM+S(I,J)*(-1.)
    2 B(I)=SUM
      B(6)=1.0
      X=B(2)/(B(1)-B(3))
      T=ATAN(X)/2.
      BP(1)=B(1)*(COS(T))**2.+B(2)*(SIN(T)*COS(T))+B(3)*(SIN(T))**2.
      BP(2)=0.
      BP(3)=B(1)*(SIN(T))**2-B(2)*SIN(T)*COS(T)+B(3)*(COS(T))**2.
      BP(4)=B(4)*COS(T)+B(5)*SIN(T)
      BP(5)=B(5)*COS(T)-B(4)*SIN(T)
      BP(6)=B(6)
      THETA=-T
      D=(2.*B(1)*COS(T)+B(2)*SIN(T))*(2.*B(3)*COS(T)-B(2)*SIN(T))-(-2.*B
     1(1)*SIN(T)+B(2)*COS(T))*(2.*B(3)*SIN(T)+B(2)*COS(T))
      PTS(1)=((-B(4)*COS(T)-B(5)*SIN(T))*(2.*B(3)*COS(T)-B(2)*SIN(T))-(B
     1(4)*SIN(T)-B(5)*COS(T))*(2.*B(3)*SIN(T)+B(2)*COS(T)))/D
      PTS(2)=((2.*B(1)*COS(T)+B(2)*SIN(T))*(B(4)*SIN(T)-B(5)*COS(T))-(-2
     1.*B(1)*SIN(T)+B(2)*COS(T))*(-B(4)*COS(T)-B(5)*SIN(T)))/D
      Y=BP(3)*(BP(4)**2)+BP(1)*(BP(5)**2)-4.*BP(1)*BP(3)*BP(6)
      F=ABS(B(1))
      G=ABS(B(3))
      IF(F-G) 20,20,21
   20 AXIS(1)=Y/(4.*(BP(1)**2)*BP(3))
      AXIS(2)=Y/(4.*(BP(3)**2)*BP(1))
      GO TO 24
   21 AXIS(2)=Y/(4.*(BP(1)**2)*BP(3))
      AXIS(1)=Y/(4.*(BP(3)**2)*BP(1))
   24 AXIS(1)=SQRT(AXIS(1))
      AXIS(2)=SQRT(AXIS(2))
      RETURN
      END
```

NASA—MSFC